

Dependently Typed Programming: an Agda introduction

Conor McBride

February 7, 2011

Chapter 1

Vectors and Finite Sets

```
data List (X : Set) : Set where
```

```
⟨⟩ : List X
_>_ : X → List X → List X
```

```
zap : {S T : Set} → List (S → T) → List S → List T
```

```
zap ⟨⟩ ⟨⟩ = ⟨⟩
```

```
zap (f, fs) (s, ss) = f s, zap fs ss
```

```
zap _ _ = ⟨⟩ -- a dummy value, for cases we should not reach
```

That's the usual 'garbage in? garbage out!' deal. Logically, we might want to ensure the inverse: if we supply meaningful input, we want meaningful output. But what is meaningful input? Lists the same length! Locally, we have a *relative* notion of meaningfulness. What is meaningful output? We could say that if the inputs were the same length, we expect output of that length. How shall we express this property?

```
data Nat : Set where
```

```
zero : Nat
```

```
suc : Nat → Nat
```

```
length : {X : Set} → List X → Nat
```

```
length ⟨⟩ = zero
```

```
length (x, xs) = suc (length xs)
```

Informally,¹ we might state and prove something like

$$\forall fs, ss. \text{length } fs = \text{length } ss \Rightarrow \text{length } (\text{zap } fs \text{ } ss) = \text{length } fs$$

by structural induction [Burstall, 1969] on fs , say. Of course, we could just as well have concluded that $\text{length } (\text{zap } fs \text{ } ss) = \text{length } ss$, and if we carry on *zapping*, we shall accumulate a multitude of expressions known to denote the same number.

What can we say about list concatenation? We may define addition.

```
_+_N_ : Nat → Nat → Nat
```

```
zero +_N y = y
```

```
suc x +_N y = suc (x +_N y)
```

Agda has a very simple lexer and very few special characters. To a first approximation, `(){};` stand alone and everything else must be delimited with whitespace.

The number of c's in `suc` is a long standing area of open warfare.

Agda users tend to use lowercase-vs-uppercase to distinguish things in `Sets` from things which are or manipulate `Sets`.

How many ways to define `+N`?

¹by which I mean, not to a computer

We may define concatenation.

```


$$\begin{aligned}
& \_+_{\mathbf{L}}\_ : \{X : \mathbf{Set}\} \rightarrow \mathbf{List} \, X \rightarrow \mathbf{List} \, X \rightarrow \mathbf{List} \, X \\
& \langle \rangle \quad \_+_{\mathbf{L}} \, ys = ys \\
& (x, xs) \_+_{\mathbf{L}} \, ys = x, (xs \_+_{\mathbf{L}} \, ys)
\end{aligned}$$


```

It takes a proof by induction (and a convenient definition of $_+_{\mathbf{N}}$) to note that

$$\mathbf{length} \, (xs _+_{\mathbf{L}} \, ys) = \mathbf{length} \, xs _+_{\mathbf{N}} \, \mathbf{length} \, ys$$

Matters get worse if we try to work with matrices as lists of lists (a matrix is a column of rows, say). How do we express rectangularity? Can we define a function to compute the dimensions of a matrix? Do we want to? What happens in degenerate cases? Given m, n , we might at least say that the outer list has length m and that all the inner lists have length n . Talking about matrices gets easier if we imagine that the dimensions are *prescribed*—to be checked, not measured.

1.0.1 Peano Exercises

Exercise 1.1 (Go Forth and Multiply!) *Given addition, implement multiplication.*

```


$$\_ \times_{\mathbf{N}} \_ : \mathbf{Nat} \rightarrow \mathbf{Nat} \rightarrow \mathbf{Nat}$$


```

Exercise 1.2 (Subtract with Dummy) *Implement subtraction, with a nasty old dummy return when you take a big number from a small one.*

```


$$\_ -_{\mathbf{N}} \_ : \mathbf{Nat} \rightarrow \mathbf{Nat} \rightarrow \mathbf{Nat}$$


```

Exercise 1.3 (Divide with a Duplicate) *Implement division. Agda won't let you do repeated subtraction directly (not structurally decreasing), but you can do something sensible (modulo the dummy) like this:*

```


$$\begin{aligned}
& \_ \div_{\mathbf{N}} \_ : \mathbf{Nat} \rightarrow \mathbf{Nat} \rightarrow \mathbf{Nat} \\
& x \div_{\mathbf{N}} d = \mathbf{help} \, x \, d \, \mathbf{where} \\
& \mathbf{help} : \mathbf{Nat} \rightarrow \mathbf{Nat} \rightarrow \mathbf{Nat} \\
& \mathbf{help} \, x \, e = \_ \{!!\}
\end{aligned}$$


```

You can recursively peel \mathbf{sucs} from e one at a time, with the original d still in scope.

1.1 Vectors

Here are lists, indexed by numbers which happen to measure their length: these are known in the trade as *vectors*.

Agda allows overloading of constructors, as its approach to typechecking is of a bidirectional character

Might want to say something about head and tail, and about how coverage checking works anyway.

Not greatly enamoured of $S \, T : \mathbf{Set}$ notation, but there it is.

`vec` is an example of

```


$$\begin{aligned}
& \mathbf{data} \, \mathbf{Vec} \, (X : \mathbf{Set}) : \mathbf{Nat} \rightarrow \mathbf{Set} \, \mathbf{where} \\
& \langle \rangle : \mathbf{Vec} \, X \, \mathbf{zero} \\
& \_ \rightarrow \_ : \{n : \mathbf{Nat}\} \rightarrow X \rightarrow \mathbf{Vec} \, X \, n \rightarrow \mathbf{Vec} \, X \, (\mathbf{suc} \, n) \\
& \mathbf{vap} : \{n : \mathbf{Nat}\} \{S \, T : \mathbf{Set}\} \rightarrow \mathbf{Vec} \, (S \rightarrow T) \, n \rightarrow \mathbf{Vec} \, S \, n \rightarrow \mathbf{Vec} \, T \, n \\
& \mathbf{vap} \, \langle \rangle \quad \langle \rangle = \langle \rangle \\
& \mathbf{vap} \, (f, fs) \, (s, ss) = f \, s, \mathbf{vap} \, fs \, ss
\end{aligned}$$


```

```

vec : { n : Nat } { X : Set } → X → Vec X n
vec { zero } x = ⟨ ⟩
vec { suc n } x = x, vec x

```

```

_+v+_ : { m n : Nat } { X : Set } → Vec X m → Vec X n → Vec X (m +N n)
⟨ ⟩ +v+ ys = ys
(x, xs) +v+ ys = x, (xs +v+ ys)

```

By now, you may have noticed the proliferation of listy types.

```

vrevapp : { m n : Nat } { X : Set } → Vec X m → Vec X n → Vec X (m +N n)
vrevapp ⟨ ⟩ ys = ys
vrevapp (x, xs) ys = -- | {! vrevapp xs (x, ys) !} |

```

Here's a stinker. Of course, you can rejig vrevapp to be tail recursive and make +v+ a stinker.

Which other things work badly? Filter?

I wanted to make _/_ left-associative, but no such luck.

```

vtraverse : { F : Set → Set } →
  ({ X : Set } → X → F X) →
  ({ S T : Set } → F (S → T) → F S → F T) →
  { n : Nat } { X Y : Set } →
  (X → F Y) → Vec X n → F (Vec Y n)
vtraverse pure _/_ f ⟨ ⟩ = pure ⟨ ⟩
vtraverse pure _/_ f (x, xs) = (pure _, _ / f x) / vtraverse pure _/_ f xs

```

When would be a good time to talk about universe polymorphism?

```

ι : { X : Set } → X → X
ι x = x
κ : { X Y : Set } → X → Y → X
κ x y = x

```

Why is Y undetermined?

```

vsum : { n : Nat } → Vec Nat n → Nat
vsum = vtraverse (κ zero) _+N_ { Y = Nat } ι

```

1.1.1 Matrix Exercises

Let us define an m by n matrix to be a vector of m rows, each length n .

```

Matrix : Nat → Nat → Set → Set
Matrix m n X = Vec (Vec X n) m

```

Exercise 1.4 (Matrices are Applicative) Show that $\text{Matrix } m \ n$ can be equipped with operations analogous to vec and vap .

```

vvec : { m n : Nat } { X : Set } → X → Matrix m n X
vvap : { m n : Nat } { S T : Set } →
  Matrix m n (S → T) → Matrix m n S → Matrix m n T

```

which, respectively, copy a given element into each position, and apply functions to arguments in corresponding positions.

Exercise 1.5 (Matrix Addition) Use the applicative interface for Matrix to define their elementwise addition.

```

_+M_ : { m n : Nat } → Matrix m n Nat → Matrix m n Nat → Matrix m n Nat

```

Exercise 1.6 (Matrix Transposition) Use `vtraverse` to give a one-line definition of matrix transposition.

```
transpose : {m n : Nat} {X : Set} → Matrix m n X → Matrix n m X
```

Exercise 1.7 (Identity Matrix) Define a function

```
idMatrix : {n : Nat} → Matrix n n Nat
```

Exercise 1.8 (Matrix Multiplication) Define matrix multiplication. There are lots of ways to do this. Some involve defining scalar product, first.

```
_×M_ : {l m n : Nat} → Matrix l m Nat → Matrix m n Nat → Matrix l n Nat
```

1.1.2 Unit and Sigma types

Why do this with records?

```
record 1 : Set where
  constructor ⟨⟩
  open 1 public
```

The `field` keyword declares fields, we can also add ‘manifest’ fields.

```
record Σ (S : Set) (T : S → Set) : Set where
  constructor →, -
  field
    fst : S
    snd : T fst
  open Σ public
  _×_ : Set → Set → Set
  S × T = Σ S λ _ → T
```

1.1.3 Apocrypha

You would not invent dependent pattern matching if vectors were your only example.

```
VecR : Set → Nat → Set
VecR X zero = 1
VecR X (suc n) = X × VecR X n
```

The definition is logically the same, why are the programs noisier?

```
vconcr : {m n : Nat} {X : Set} →
  VecR X m → VecR X n → VecR X (m +N n)
vconcr {zero} ⟨⟩ ys = ys
vconcr {suc m} (x, xs) ys = x, vconcr {m} xs ys
```

```
data ==_ {X : Set} (x : X) : X → Set where
  ⟨⟩ : x == x
```

```
len : {X : Set} → List X → Nat
len ⟨⟩ = zero
len (x, xs) = suc (len xs)
```

Agda’s `λ` scopes rightward as far as possible, reducing bracketing. Even newer fancy binding sugar might

```
VecP : Set → Nat → Set
VecP X n = Σ (List X) λ xs → len xs = n
```

```
vnil : {X : Set} → VecP X zero
vnil = ⟨⟩, ⟨⟩
```

```
vcons : {X : Set} {n : Nat} → X → VecP X n → VecP X (suc n)
vcons x (xs, p) = (x, xs), --{!!}
```

```
vapP : {n : Nat} {S T : Set} →
      VecP (S → T) n → VecP S n → VecP T n
vapP (⟨⟩, ⟨⟩) (⟨⟩, ⟨⟩) = ⟨⟩, ⟨⟩
vapP ((f, fs), ⟨⟩) ((s, ss), p) = (f s, vap (fs, ?) (ss, ?)), ?
```

It's already getting bad here, but we can match p against $\langle \rangle$ and complete.

But this really is toxic.

1.2 Finite Sets

If we know the size of a vector, can we hope to project from it safely? Here's a family of *finite sets*, good to use as indices into vectors.

```
data Fin : Nat → Set where
  zero : {n : Nat} → Fin (suc n)
  suc : {n : Nat} → (i : Fin n) → Fin (suc n)
```

Finite sets are sets of bounded numbers. One thing we may readily do is forget the bound.

```
fog : {n : Nat} → Fin n → Nat
fog zero = zero
fog (suc i) = suc (fog i)
```

Do you resent writing this function? You should.

Now let's show how to give a total projection from a vector of known size.

```
vproj : {n : Nat} {X : Set} → Vec X n → Fin n → X
vproj ⟨⟩ () = ()
vproj (x, xs) zero = x
vproj (x, xs) (suc i) = vproj xs i
```

Here's our first Aunt Fanny. We could also swap the arguments around.

Suppose we want to project at an index not known to be suitably bounded. How might we check the bound? We shall return to that thought, later.

It's always possible to give enough Aunt Fannies to satisfy the coverage checker.

1.2.1 Renamings

We'll shortly use `Fin` to type bounded sets of de Bruijn indices. Functions from one finite set to another will act as 'renamings'.

Extending the context with a new assumption is sometimes known as 'weakening': making more assumptions weakens an argument. Suppose we have a function from `Fin m` to `Fin n`, renaming variables, as it were. How should weakening act on this function? Can we extend the function to the sets one larger, mapping the 'new' source zero to the 'new' target zero? This operation shows how to push a renaming under a binder.

Categorists, what should we prove about `weaken`?

```

weaken : { m n : Nat } → (Fin m → Fin n) → Fin (suc m) → Fin (suc n)
weaken f zero    = zero
weaken f (suc i) = suc (f i)

```

One operation we'll need corresponds to inserting a new variable somewhere in the context. This operation is known as 'thinning'. Let's define the order-preserving injection from $\text{Fin } n$ to $\text{Fin } (\text{suc } n)$ which misses a given element

```

thin : { n : Nat } → Fin (suc n) → Fin n → Fin (suc n)
thin      zero    = suc
thin { zero } (suc ())
thin { suc n } (suc i) = weaken (thin i)

```

1.2.2 Finite Set Exercises

Exercise 1.9 (Tabulation) Invert vproj . Given a function from a Fin set, show how to construct the vector which tabulates it.

```

vtab : { n : Nat } { X : Set } → (Fin n → X) → Vec X n

```

Exercise 1.10 (Plan a Vector) Show how to construct the 'plan' of a vector—a vector whose elements each give their own position, counting up from zero .

```

vplan : { n : Nat } → Vec (Fin n) n

```

Exercise 1.11 (Max a Fin) Every nonempty finite set has a smallest element zero and a largest element which has as many sucs as allowed. Construct the latter

```

max : { n : Nat } → Fin (suc n)

```

Exercise 1.12 (Embed, Preserving fog) Give the embedding from one finite set to the next which preserves the numerical value given by fog .

```

emb : { n : Nat } → Fin n → Fin (suc n)

```

Exercise 1.13 (Thickening) Construct $\text{thick } i$ the partial inverse of $\text{thin } i$. You'll need

```

data Maybe (X : Set) : Set where
  yes : X → Maybe X
  no  :      Maybe X

```

Which operations on Maybe will help? Discover and define them as you implement:

```

thick : { n : Nat } → Fin (suc n) → Fin (suc n) → Maybe (Fin n)

```

Note that thick acts as an inequality test.

Exercise 1.14 (Order-Preserving Injections) Define an inductive family

```

OPI : Nat → Nat → Set

```

such that $\text{OPI } m \ n$ gives a unique first-order representation to exactly the order-preserving injections from $\text{Fin } m$ to $\text{Fin } n$, and give the functional interpretation of your data. Show that OPI is closed under identity and composition.

Chapter 2

Lambda Calculus with de Bruijn Indices

I'm revisiting chapter 7 of my thesis here.

Here are the λ -terms with n available de Bruijn indices [de Bruijn, 1972].

```
data Tm (n : Nat) : Set where
  var  : Fin n → Tm n
  $    : Tm n → Tm n → Tm n
  lam  : Tm (suc n) → Tm n
infixl 6 $
```

Which operations work?

Substitute for **zero**?

```
sub0 : {n : Nat} → Tm n → Tm (suc n) → Tm n
sub0 s (var zero)    = s
sub0 s (var (suc i)) = var i
sub0 s (f $ a)       = sub0 s f $ sub0 s a
sub0 s (lam b)       = lam (sub0 ? b)
```

How many different
kinds of trouble are
we in?

Simultaneous substitution?

```
ssub : {m n : Nat} → (Fin m → Tm n) → Tm m → Tm n
ssub σ (var i) = σ i
ssub σ (f $ a) = ssub σ f $ ssub σ a
ssub {m} {n} σ (lam b) = lam (ssub σ b) where
  σ : Fin (suc m) → Tm (suc n)
  σ zero    = var zero
  σ (suc i) = ssub (λ i → var (suc i)) (σ i)
```

Notoriously not
structurally recursive.

At this point, Thorsten Altenkirch and Bernhard Reus [Altenkirch and Reus, 1999] reached for the hammer of wellordering, but there's a cheaper way to get out of the jam.

2.1 Simultaneous Renaming and Substitution

You can define simultaneous renaming really easily.

```
wkr : {m n : Nat} → (Fin m → Fin n) → Fin (suc m) → Fin (suc n)
wkr ρ zero    = zero
```

```

wkr  $\rho$  (suc  $i$ ) = suc ( $\rho$   $i$ )
ren : {  $m$   $n$  : Nat }  $\rightarrow$  (Fin  $m \rightarrow$  Fin  $n$ )  $\rightarrow$  Tm  $m \rightarrow$  Tm  $n$ 
ren  $\rho$  (var  $i$ ) = var ( $\rho$   $i$ )
ren  $\rho$  ( $f$  $  $a$ ) = ren  $\rho$   $f$  $ ren  $\rho$   $a$ 
ren  $\rho$  (lam  $b$ ) = lam (ren (wkr  $\rho$ )  $b$ )

```

And you can define substitution, given renaming.

```

wks : {  $m$   $n$  : Nat }  $\rightarrow$  (Fin  $m \rightarrow$  Tm  $n$ )  $\rightarrow$  Fin (suc  $m$ )  $\rightarrow$  Tm (suc  $n$ )
wks  $\sigma$  zero = var zero
wks  $\sigma$  (suc  $i$ ) = ren suc ( $\sigma$   $i$ )
sub : {  $m$   $n$  : Nat }  $\rightarrow$  (Fin  $m \rightarrow$  Tm  $n$ )  $\rightarrow$  Tm  $m \rightarrow$  Tm  $n$ 
sub  $\sigma$  (var  $i$ ) =  $\sigma$   $i$ 
sub  $\sigma$  ( $f$  $  $a$ ) = sub  $\sigma$   $f$  $ sub  $\sigma$   $a$ 
sub  $\sigma$  (lam  $b$ ) = lam (sub (wks  $\sigma$ )  $b$ )

```

How repetitive! Let's abstract out the pattern.

```

record Kit ( $I$  : Nat  $\rightarrow$  Set) : Set where
  constructor mkKit
  field
    mkv : {  $n$  : Nat }  $\rightarrow$  Fin  $n \rightarrow$   $I$   $n$ 
    mkt : {  $n$  : Nat }  $\rightarrow$   $I$   $n \rightarrow$  Tm  $n$ 
    wki : {  $n$  : Nat }  $\rightarrow$   $I$   $n \rightarrow$   $I$  (suc  $n$ )
  open Kit public

wk : {  $I$  : Nat  $\rightarrow$  Set }  $\rightarrow$  Kit  $I \rightarrow$  {  $m$   $n$  : Nat }  $\rightarrow$ 
  (Fin  $m \rightarrow$   $I$   $n$ )  $\rightarrow$  Fin (suc  $m$ )  $\rightarrow$   $I$  (suc  $n$ )
wk  $k$   $\tau$  zero = mkv  $k$  zero
wk  $k$   $\tau$  (suc  $i$ ) = wki  $k$  ( $\tau$   $i$ )

act : {  $I$  : Nat  $\rightarrow$  Set }  $\rightarrow$  Kit  $I \rightarrow$  {  $m$   $n$  : Nat }  $\rightarrow$ 
  (Fin  $m \rightarrow$   $I$   $n$ )  $\rightarrow$  Tm  $m \rightarrow$  Tm  $n$ 
act  $k$   $\tau$  (var  $i$ ) = mkt  $k$  ( $\tau$   $i$ )
act  $k$   $\tau$  ( $f$  $  $a$ ) = act  $k$   $\tau$   $f$  $ act  $k$   $\tau$   $a$ 
act  $k$   $\tau$  (lam  $b$ ) = lam (act  $k$  (wk  $k$   $\tau$ )  $b$ )

```

2.1.1 Exercises

Exercise 2.1 (Renaming Kit) Define the renaming kit.

```
renk : Kit Fin
```

Exercise 2.2 (Substitution Kit) Define the substitution kit.

```
subk : Kit Tm
```

Exercise 2.3 (Substitute zero) `sub0` : { n : Nat } \rightarrow Tm $n \rightarrow$ Tm (suc n) \rightarrow Tm n

Exercise 2.4 (Reduce One) Define a function to contract the leftmost redex in a λ -term, if there is one.

```
leftRed : {  $n$  : Nat }  $\rightarrow$  Tm  $n \rightarrow$  Maybe (Tm  $n$ )
```

Exercise 2.5 (Complete Development) Show how to compute the complete development of a λ -term, contracting all its visible redexes in parallel (but not the redexes which thus arise).

`develop` : $\{n : \text{Nat}\} \rightarrow \text{Tm } n \rightarrow \text{Tm } n$

Exercise 2.6 (Gasoline Alley) Write an iterator, computing the n -fold self-composition of an endofunction, effectively interpreting each `Nat` as its corresponding Church numeral.

`iterate` : $\text{Nat} \rightarrow \{X : \text{Set}\} \rightarrow (X \rightarrow X) \rightarrow X \rightarrow X$

You can use `iterate` and `develop` to run λ -terms for as many steps as you like, as long as you are modest in your likes.

Exercise 2.7 (Another Substitution Recipe) It occurred to me at time of writing that one might cook substitution differently. Using abacus-style addition

`-+a-` : $\text{Nat} \rightarrow \text{Nat} \rightarrow \text{Nat}$
`zero` `+a` $n = n$
`suc` m `+a` $n = m$ `+a` `suc` n

let

`Sub` : $\text{Nat} \rightarrow \text{Nat} \rightarrow \text{Set}$
`Sub` m $n = (w : \text{Nat}) \rightarrow \text{Fin } (w$ `+a` $m) \rightarrow \text{Tm } (w$ `+a` $n)$

be the type of substitutions which can be weakened. Define

`subw` : $\{m\ n : \text{Nat}\} \rightarrow \text{Sub } m\ n \rightarrow \text{Tm } m \rightarrow \text{Tm } n$

Now show how to turn a renaming into a `Sub`.

`renSub` : $\{m\ n : \text{Nat}\} \rightarrow (\text{Fin } m \rightarrow \text{Fin } n) \rightarrow \text{Sub } m\ n$

Finally, show how to turn a simultaneous substitution into a `Sub`.

`subSub` : $\{m\ n : \text{Nat}\} \rightarrow (\text{Fin } m \rightarrow \text{Tm } n) \rightarrow \text{Sub } m\ n$

2.1.2 Simply Typed Lambda Calculus

Altenkirch and Reus carry on to develop simultaneous type-preserving substitution for the *simply-typed* λ -calculus. Let's see how.

`infixr` 4 `▷-`
`infixr` 3 `⊢-`
`infixr` 3 `⊢⊥`
`infixr` 4 `→, -`

`data` `Ty` : `Set` where
`⊢` : `Ty`
`▷-` : $(S\ T : \text{Ty}) \rightarrow \text{Ty}$

`data` `Context` : `Set` where
`⟨⟩` : `Context`
`→, -` : $(G : \text{Context}) (S : \text{Ty}) \rightarrow \text{Context}$

```

data  $\vdash$  : Context  $\rightarrow$  Ty  $\rightarrow$  Set where
  zero :  $\forall \{ G T \} \rightarrow G, T \vdash T$ 
  suc :  $\forall \{ G S T \} (x : G \vdash T) \rightarrow G, S \vdash T$ 

data  $\vdash$  : Context  $\rightarrow$  Ty  $\rightarrow$  Set where
  var :  $\forall \{ G T \} (x : G \vdash T)$ 
     $\rightarrow$  --
     $G \vdash T$ 
    --  $\lambda$ -abstraction extends the context
  lam :  $\forall \{ G S T \} (b : G, S \vdash T)$ 
     $\rightarrow$  --
     $G \vdash S \triangleright T$ 
    -- application demands a type coincidence
  $ :  $\forall \{ G S T \} (f : G \vdash S \triangleright T) (s : G \vdash S)$ 
     $\rightarrow$  --
     $G \vdash T$ 

 $\llbracket - \rrbracket_{\tau}$  : Ty  $\rightarrow$  Set
 $\llbracket \iota \rrbracket_{\tau} = \mathbf{Nat}$ 
 $\llbracket S \triangleright T \rrbracket_{\tau} = \llbracket S \rrbracket_{\tau} \rightarrow \llbracket T \rrbracket_{\tau}$ 

 $\llbracket - \rrbracket_{\mathbf{c}}$  : Context  $\rightarrow$  Set
 $\llbracket \langle \rangle \rrbracket_{\mathbf{c}} = \mathbf{1}$ 
 $\llbracket G, S \rrbracket_{\mathbf{c}} = \llbracket G \rrbracket_{\mathbf{c}} \times \llbracket S \rrbracket_{\tau}$ 
 $\llbracket - \rrbracket_{\mathbf{v}}$  :  $\forall \{ G T \} \rightarrow G \vdash T \rightarrow \llbracket G \rrbracket_{\mathbf{c}} \rightarrow \llbracket T \rrbracket_{\tau}$ 
 $\llbracket \mathbf{zero} \rrbracket_{\mathbf{v}} (\_, t) = t$ 
 $\llbracket \mathbf{suc } i \rrbracket_{\mathbf{v}} (g, \_) = \llbracket i \rrbracket_{\mathbf{v}} g$ 
 $\llbracket - \rrbracket_{\mathbf{t}}$  :  $\forall \{ G T \} \rightarrow G \vdash T \rightarrow \llbracket G \rrbracket_{\mathbf{c}} \rightarrow \llbracket T \rrbracket_{\tau}$ 
 $\llbracket \mathbf{var } x \rrbracket_{\mathbf{t}} = \llbracket x \rrbracket_{\mathbf{v}}$ 
 $\llbracket \mathbf{lam } b \rrbracket_{\mathbf{t}} = \lambda g s \rightarrow \llbracket b \rrbracket_{\mathbf{t}} (g, s)$ 
 $\llbracket f \$ s \rrbracket_{\mathbf{t}} = \lambda g \rightarrow \llbracket f \rrbracket_{\mathbf{t}} g (\llbracket s \rrbracket_{\mathbf{t}} g)$ 
eval :  $\forall \{ T \} \rightarrow \langle \rangle \vdash T \rightarrow \llbracket T \rrbracket_{\tau}$ 
eval  $t = \llbracket t \rrbracket_{\mathbf{t}} \langle \rangle$ 

example :  $\langle \rangle \vdash \_$ 
example = (lam (var zero)) $ lam (var zero)

```

Chapter 3

Views

```
data -Bounded?_ (u : Nat) : Nat → Set where
  yes : (i : Fin u) → u -Bounded? (fog i)
  no  : (x : Nat) → u -Bounded? (u +N x)
-bounded?_ : (u n : Nat) → u -Bounded? n
zero -bounded? n = no n
(suc u) -bounded? zero = yes zero
(suc u) -bounded? (suc n) with u -bounded? n
(suc u) -bounded? (suc .(fog i)) | yes i = yes (suc i)
(suc u) -bounded? (suc .(u +N x)) | no x = no x
```


Bibliography

Thorsten Altenkirch and Bernhard Reus. Monadic presentations of lambda terms using generalized inductive types. In Jörg Flum and Mario Rodríguez-Artalejo, editors, *CSL*, volume 1683 of *LNCS*, pages 453–468. Springer, 1999.

Rod Burstall. Proving properties of programs by structural induction. *Computer Journal*, 12(1):41–48, 1969.

Nicolas G. de Bruijn. Lambda Calculus notation with nameless dummies: a tool for automatic formula manipulation. *Indagationes Mathematicæ*, 34:381–392, 1972.